Protocol Audit Report

Version 1.0

ZenoraSec

July 29, 2024

Protocol Audit Report

ZenoraSec

July 29, 2024

Lead Security Researcher: ZenoraSec Jeremy Zenora

Table of Contents

Protocol Summary

Protocol handles strorage and retrieval of user's password. Protocol is meant for one single user. Only the owner of the contract (single user) can change and access the password.

Disclaimer

The ZenoraSec team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

Scope

```
./src/
#-- PoolFactory.sol
#-- TSwapPool.sol
```

Executive Summary

We found a few major issues that contradicted the logic and original use of the protocol. We recommend to rethink the way the whole protocol works as suggested in the findings below. We used foundry framework with tools such as fuzzing, symbolic/formal verification, manual review.

Issues found

Severity	Number of issues found
High	5
Medium	1
Low	2

Severity	Number of issues found
Info	6
Total	14

Findings

High

[H-1] Typo in function `TSwapPool::getInputAmountBasedOnOutput` on line `((inputReserves * outputAmount) * 10000)`, number `10_000` causes the swap fee to be 91.3%! Significantly higher than the intented 0.3% swap fee

Description: There is an additional zero added to the number `10_000`, instead of `1_000`:

```
return ((inputReserves * outputAmount) * 10000) / ((outputReserves -
    outputAmount) * 997);
```

This makes the fee unbearable for the users.

Impact: The fee becomes `91.3%` instead of `0.3%`

Recommended Mitigation: Remove one `0` from the number, use constant variables to express this number, it is not good to use `magic numbers` in the codebase.

```
-         return((inputReserves * outputAmount) * 10000) / ((
    outputReserves - outputAmount) * 997);
+         return((inputReserves * outputAmount) * 1000) / ((
    outputReserves - outputAmount) * 997);
```

[H-2] In function `TSwapPool::_swap` after 10 swaps, protocol adds 1 additional token, this breaks the invariant `x * y = k`

Description: If you add additional 1 token to the user after 10 swaps, you break the math of the protocol as described here: - `x` - The Balance of the pool token - `y` - The Balance of the WETH - `k` - The constant product of the two balances

Therefore whenever the balances change in the protocol, constant `k` should be ideal ratio between them, however this breaks once you add extra 1 token in the `_swap` function. Over time the protocol funds will be drained.

```
// x * y = k
// numberOfWeth * numberOfPoolTokens = constant k
// k must not change during a transaction (invariant)
```

this is the malicious code block:

```
if (swap_count >= SWAP_COUNT_MAX) {
            swap_count = 0;
            outputToken.safeTransfer(msg.sender, 1
                _000_000_000_000_000_000);
        }
```

Impact: This will break the invarinat k and drain funds of the contract over time, malicious user could swap many times and get very large amount of tokens.

Proof of Concept: 1. User makes 10 swaps 2. User gets additional token and repeats this until funds are gone from the protocol. 3. protocol invariant k changes and breaks.

PoC

```
function testInvariantBreaks() public {
        //we add liquidity
        vm.startPrank(liquidityProvider);
        poolToken.approve(address(pool), type(uint256).max);
        weth.approve(address(pool), type(uint256).max);
        pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
        vm.stopPrank();
        uint256 outputWeth = 1e17;
        vm.startPrank(user);
        poolToken.approve(address(pool), type(uint256).max);
        //Lets make 10 swaps
        pool.swapExactOutput(
            poolToken,
            weth,
            outputWeth,
            uint64(block.timestamp)
        );
        pool.swapExactOutput(
            poolToken,
            weth,
            outputWeth,
            uint64(block.timestamp)
        );
        pool.swapExactOutput(
            poolToken,
            weth,
            outputWeth,
            uint64(block.timestamp)
        );
        pool.swapExactOutput(
```

```
            poolToken,
            weth,
            outputWeth,
            uint64(block.timestamp)
        );
        pool.swapExactOutput(
            poolToken,
            weth,
            outputWeth,
            uint64(block.timestamp)
        );
        pool.swapExactOutput(
            poolToken,
            weth,
            outputWeth,
            uint64(block.timestamp)
        );
        pool.swapExactOutput(
            poolToken,
            weth,
            outputWeth,
            uint64(block.timestamp)
        );
        pool.swapExactOutput(
            poolToken,
            weth,
            outputWeth,
            uint64(block.timestamp)
        );
        pool.swapExactOutput(
            poolToken,
            weth,
            outputWeth,
            uint64(block.timestamp)
        );
        pool.swapExactOutput(
            poolToken,
            weth,
            outputWeth,
            uint64(block.timestamp)
        );

        int256 startingY = int256(weth.balanceOf(address(pool)));
        int256 expectedDeltaY = int256(-1) * int256(outputWeth);

        pool.swapExactOutput(
            poolToken,
            weth,
            outputWeth,
            uint64(block.timestamp)
        );

        vm.stopPrank();

        uint256 endingY = weth.balanceOf(address(pool));
        int256 actualDeltaY = int256(endingY) - int256(startingY);
        assertEq(actualDeltaY, expectedDeltaY);
```

```
82      }
```

Recommended Mitigation: Remove the functionality of sending 1 token to the user, or find a different way to reward your users, for example a `claim` function where users can claim reward if they did enough transactions.

[H-3] Function `TSwapPool::sellPoolTokens` mismatches parameters when calling the `TSwapPool::swapExactOutput`, resulting in users receiving incorrect amount of WETH.

Description: In function `sellPoolTokens` users try to sell their exact pool token amount as input for weth, then function `swapExactOutput` gets called where we take outputAmount as parameter, for this reason we should call `TSwapPool::swapExactInput` if we want to sell exact input amount.

Impact: Users sell incorrect amount of tokens, which is severe disruption of the protocol functionality.

Proof of Concept: 1. User has 100e18 poolTokens. 2. User tries to sell ONLY 5 poolTokens using current `sellPoolTokens`. 3. User ends up selling completely different poolToken amount.

Output from my PoC below:

```
Logs:
  BEFORE: poolTokens of user:    100
  BEFORE: weth balance of user:  100
  AFTER: poolTokens of user:     47
  AFTER: weth balance of user:   105
```

PoC

```
function testSellPoolTokensAmount() public {
        //we add liquidity
        vm.startPrank(liquidityProvider);
        poolToken.approve(address(pool), type(uint256).max);
        weth.approve(address(pool), type(uint256).max);
        pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
        vm.stopPrank();

        console.log(
            "BEFORE: poolTokens of user:   ",
            poolToken.balanceOf(user) / 1e18
        );
        console.log(
            "BEFORE: weth balance of user: ",
            weth.balanceOf(user) / 1e18
        );
        vm.startPrank(user);
        poolToken.approve(address(pool), type(uint256).max);
```

```
19          weth.approve(address(pool), type(uint256).max);
20
21          //We try to sell ONLY 5 pool tokens
22          pool.sellPoolTokens(5e18);
23          vm.stopPrank();
24
25          console.log(
26              "AFTER: poolTokens of user:    ",
27              poolToken.balanceOf(user) / 1e18
28          );
29          console.log(
30              "AFTER: weth balance of user:  ",
31              weth.balanceOf(user) / 1e18
32          );
33      }
```

Recommended Mitigation: To fix this, simply call `swapExactInput` instead of `swapExactOutput`. And add additional parameter to be passed to your `sellPoolTokens`, e.g. `uint256 minOutputAmount`.

```
1  function sellPoolTokens(
2  -       uint256 poolTokenAmount
3  +       uint256 poolTokenAmount, uint256 minOutputAmount
4      ) external returns (uint256 wethAmount) {
5          return
6  -           swapExactOutput(
7  -               i_poolToken,
8  -               i_wethToken,
9  -               //@reported-high this is wrong, use swapexactinput
10 -               poolTokenAmount,
11 -               uint64(block.timestamp)
12
13 +           swapExactInput(
14 +               i_poolToken,
15 +               poolTokenAmount,
16 +               i_wethToken,
17 +               minOutputAmount,
18 +               uint64(block.timestamp)
19             );
20     }
```

[H-4] Lack of slippage protection in `TSwapPool::swapExactOutput`, causing potentionally significant losses for users that swap

Description: in `swapExactOutput` you do not have any checks for slippage. This function is similar to `TSwapPool::swapExactInput` where you do check if the `outputAmount` is less than `minOutputAmount` which prevents from slipapge attacks. You should specify `maxInputAmount`

in the `swapExactOutput`.

Impact: If the price changes significantly during the transaction, the user can get very bad rate.

Proof of Concept: scenario not accounting for fees: 1. Price of 1 WETH is 1000 USDC 2. User swaps for 1 WETH 1. input token is USDC 2. output token is WETH 3. we want 1 weth amount 4. deadline = whatever 3. The function does not offer maxInputAmount (so we can end up paying whatever for the 1 weth) 4. the price changes, now 1 WETH costs 10,000 USDC and we still pay it. 5. the user spends 10,000 USDC for the 1 WETH, 10x more than they expected.

PoC Logs:

```
  Price of 1 WETH ON START: 0.91 /  914145202967408518 THAT USER THINKS
      HE IS BUYING AT
  Price of 1 WETH AFTER FRONTRUN:  3
  Price of 1 WETH AFTER USER BOUGHT  11
```

PoC

```
function testSlippageOnSwapExactOutput() public {
        //we initialize the attacker's balance
        address attacker = makeAddr("attacker");
        poolToken.mint(attacker, 100e18);
        weth.mint(attacker, 100e18);

        //we add liquidity using independent wallet
        vm.startPrank(liquidityProvider);
        poolToken.approve(address(pool), type(uint256).max);
        weth.approve(address(pool), type(uint256).max);
        pool.deposit(11e18, 11e18, 11e18, uint64(block.timestamp));
        vm.stopPrank();

        //checking price of 1 WETH when the user looks at the pool and
           decides to make a swap.
        uint256 priceOfOneWETHBefore = pool.
           getPriceOfOneWethInPoolTokens();

        vm.startPrank(attacker);
        poolToken.approve(address(pool), type(uint256).max);
        weth.approve(address(pool), type(uint256).max);
        pool.swapExactInput(
            poolToken,
            10e18,
            weth,
            1e18,
            uint64(block.timestamp)
        );
        vm.stopPrank();

        uint256 priceAfterAttackerFrontran = pool
```

```
30              .getPriceOfOneWethInPoolTokens();
31
32          //user victim tries to swap too, does not know the price
               increased and cannot set parameters for his desired price.
33          vm.startPrank(user);
34          poolToken.approve(address(pool), type(uint256).max);
35          weth.approve(address(pool), type(uint256).max);
36          pool.swapExactOutput(poolToken, weth, 1e18, uint64(block.
               timestamp));
37          vm.stopPrank();
38          uint256 priceAfterUserBought = pool.
               getPriceOfOneWethInPoolTokens();
39          console.log(
40              "Price of 1 WETH ON START: 0.91 or ",
41              priceOfOneWETHBefore,
42              " Because of the mistake in fee calculation, which takes
                   91%, we get 0, raw price number is 91..."
43          );
44          console.log(
45              "Price of 1 WETH AFTER FRONTRUN: ",
46              priceAfterAttackerFrontran / 1e18
47          );
48          console.log(
49              "Price of 1 WETH AFTER USER BOUGHT ",
50              priceAfterUserBought / 1e18
51          );
52      }
```

Recommended Mitigation:

Add parameter `uint256 maxInputAmount` to function, and then revert if the `inputAmount` required to buy desired output is greater than `maxInputAmount`.

```
1       function swapExactOutput(
2           IERC20 inputToken,
3   +       uint256 maxInputAmount,
4           IERC20 outputToken,
5           uint256 outputAmount,
6           uint64 deadline
7   .
8   .
9   .
10
11          inputAmount = getInputAmountBasedOnOutput(
12              outputAmount,
13              inputReserves,
14              outputReserves
15          );
16
17  +       if (inputAmount > maxInputAmount)
18  +          { revert; }
```

```
19          _swap(inputToken, inputAmount, outputToken, outputAmount);
```

[H-5] Rebase tokens, and other "unusual" tokens with fee on transfer break the invariant equation!

Description: If the TSwapPool is exposed to ERC20's with unusual functionality, like fees on transfers, it can severly disrupt its functioning. the constant k is not equal to what it should be anymore.

Impact: Protocol logic fails.

Proof of Concept:

add this import to your test suite

```
1 import {WEIRDERC} from "./mocks/ERC20WEIRDERC.sol";
```

Add this Contract to your "./mocks/" folder next to ERC20Mock.sol

WEIRDERC Contract

```
1  // SPDX-License-Identifier: MIT
2  pragma solidity 0.8.20;
3
4  import {ERC20} from "@openzeppelin/contracts/token/ERC20/ERC20.sol";
5
6  contract WEIRDERC is ERC20 {
7      uint256 public constant INITIAL_SUPPLY = 1_000_000e18;
8      address public immutable owner;
9      uint256 public constant FEE = 27; // Fee percentage
10     uint256 public constant PRECISION = 100;
11
12     event TransferWithFee(address from, address to, uint256 value,
           uint256 fee);
13
14     constructor() ERC20("WEIRDERC", "WEIRDERC") {
15         owner = msg.sender;
16         _mint(msg.sender, INITIAL_SUPPLY);
17     }
18
19     function mint(address to, uint256 amount) public {
20         _mint(to, amount);
21     }
22
23     function transfer(
24         address to,
25         uint256 amount
26     ) public virtual override returns (bool) {
27         _transferWithFee(_msgSender(), to, amount);
28         return true;
```

```
    }

    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transferWithFee(from, to, amount);
        return true;
    }

    function _transferWithFee(
        address from,
        address to,
        uint256 amount
    ) internal {
        uint256 fee = (amount * FEE) / PRECISION;
        uint256 amountAfterFee = amount - fee;
        emit TransferWithFee(from, to, amountAfterFee, fee); // Emit event for debugging
        super._transfer(from, owner, fee); // Burn the fee by sending to zero address
        super._transfer(from, to, amountAfterFee);
    }
}
```

Fee on Transfer messes up our pool, now it actual delta is not the expected delta.

```
assertion failed: 5399025358904997822 != -1000000000000000000
```

PoC

```
function testWeirdERC20Behaviour() public {
        vm.startPrank(liquidityProvider);
        WEIRDERC weirdPoolToken = new WEIRDERC();
        TSwapPool weirdPool = new TSwapPool(
            address(weirdPoolToken),
            address(weth),
            "WeirdToken",
            "WEIRD"
        );

        weirdPoolToken.mint(liquidityProvider, 100e18);
        weirdPoolToken.approve(address(weirdPool), type(uint256).max);
        weth.approve(address(weirdPool), type(uint256).max);
        weirdPool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
        vm.stopPrank();

```

```
17          uint256 outputWeth = 1e18;
18
19          vm.startPrank(user);
20          weirdPoolToken.mint(user, 100e18);
21          weirdPoolToken.approve(address(weirdPool), type(uint256).max);
22          weth.approve(address(weirdPool), type(uint256).max);
23
24          int256 startingX = int256(weirdPoolToken.balanceOf(address(
                weirdPool)));
25          int256 expectedDeltaX = int256(-1) * int256(outputWeth);
26
27          weirdPool.swapExactOutput(
28              weirdPoolToken,
29              weth,
30              outputWeth,
31              uint64(block.timestamp)
32          );
33
34          uint256 endingX = weirdPoolToken.balanceOf(address(weirdPool));
35          int256 actualDeltaX = int256(endingX) - int256(startingX);
36
37          assertEq(actualDeltaX, expectedDeltaX);
38          vm.stopPrank();
39      }
```

Recommended Mitigation: Consider restricting the protocol for use of only certain types of ERC20s, not all of them.

Medium

[M-1] In `TSwapPool::deposit` parameter `deadline` is not used, causing the transactions to complete even after specifiec deadline

Description: You ask for input parameter `deadline` which according to the natspec is: "The deadline for the transaction to be completed by". But you do not use it later on in the function to actually limit the transaction to cancel after certain period.

Impact: Transaction can go through at random time, potentially hurting the depositor.

Proof of Concept: The parameter `deadline` is unused.

Recommended Mitigation: Add modifier to prevent transaction to complete after deadline

```
1 function deposit(
2          uint256 wethToDeposit,
3          uint256 minimumLiquidityTokensToMint,
4          uint256 maximumPoolTokensToDeposit,
```

```
5           //@audit-high, parameter `deadline` is not used, users can call
                deposit function and have their tx go through even after
               specified deadline. Disruption of functionality.
6           uint64 deadline
7       )
8           external
9   +       revertIfDeadlinePassed(deadline)
10          revertIfZero(wethToDeposit)
11          returns (uint256 liquidityTokensToMint)
```

Low

[L-1] Event `TSwapPool::LiquidityAdded` is emited with parameters in incorrent order

Description: When emitting the event in the `TSwapPool::_addLiquidityMintAndTransfer` function, it logs values in wrong order. the `poolTokensToDeposit` variable should be on 3rd spot and `wethToDeposit` should be 2nd.

This is how the event is looks:

```
1 event LiquidityAdded(address indexed liquidityProvider, uint256
     wethDeposited, uint256 poolTokensDeposited);
```

Impact: This leads to incorrect off-chain data.

Recommended Mitigation: Swap the two variables

```
1 -     emit LiquidityAdded(msg.sender, poolTokensToDeposit, wethToDeposit)
      ;
2 +     emit LiquidityAdded(msg.sender, wethToDeposit, poolTokensToDeposit)
      ;
```

[L-2] Return value of `TSwapPool::swapExactInput` will always be `0` because of wrong return parameter

Description: You return `uint256 output`, but there is no value assigned to that variable in the function.

Impact: Function `TSwapPool::swapExactInput` will confuse users and return wrong value `0`.

Proof of Concept:

1. User calls the `swapExactInput` function
2. Return value logged is always `0` even with successful transactions and different values.

```
Logs:
  Return value:  0
```

PoC

```
function testReturnValueOfswapExactInput() public {
        //we add liquidity
        vm.startPrank(liquidityProvider);
        poolToken.approve(address(pool), type(uint256).max);
        weth.approve(address(pool), type(uint256).max);
        pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
        vm.stopPrank();

        vm.startPrank(user);
        poolToken.approve(address(pool), type(uint256).max);
        weth.approve(address(pool), type(uint256).max);

        uint256 returnValue = pool.swapExactInput(
            poolToken,
            1e18,
            weth,
            1e17,
            uint64(block.timestamp)
        );
        vm.stopPrank();
        console.log("Return value: ", returnValue);
    }
```

Recommended Mitigation: Either rename the return value `output` to `outputAmount` or rename the `outputAmount` to `output` in the rest of the function as below.

```
function swapExactInput(IERC20 inputToken, uint256 inputAmount, IERC20
    outputToken, uint256 minOutputAmount, uint64 deadline) public
    revertIfZero(inputAmount)
        returns (
            //@audit-low returning 0 always
            uint256 output
        ) {
        uint256 inputReserves = inputToken.balanceOf(address(this));
        uint256 outputReserves = outputToken.balanceOf(address(this));

-       uint256 outputAmount = getOutputAmountBasedOnInput(inputAmount,
     inputReserves, outputReserves);
+       uint256 output = getOutputAmountBasedOnInput(inputAmount,
    inputReserves, outputReserves);

-       if (outputAmount < minOutputAmount)
-          { revert TSwapPool__OutputTooLow(outputAmount,
    minOutputAmount); }
-       _swap(inputToken, inputAmount, outputToken, outputAmount);
```

```
15
16 +       if (output < minOutputAmount)
17 +          { revert TSwapPool__OutputTooLow(output, minOutputAmount); }
18 +       _swap(inputToken, inputAmount, outputToken, output);
19     }
```

Informational

[I-1] `PoolFactory__PoolDoesNotExist` does not get used in the code and should be removed

```
1 -   error PoolFactory__PoolDoesNotExist(address tokenAddress);
```

[I-2] Constructor in `PoolFactory` lacks a zero-check for the input address

Consider adding a `require` statement to prevent against zero addresses

```
1 constructor(address wethToken) {
2 +       require(wethToken != address(0));
3         i_wethToken = wethToken;
4     }
```

[I-3] In `PoolFactory::createPool` when assigning `liquidityTokenSymbol` you incorrectly call `.name()` instead of `.symbol()`

Change it to `.symbol()`

```
1 string memory liquidityTokenSymbol = string.concat(
2             "ts",
3 -           IERC20(tokenAddress).name()
4 +           IERC20(tokenAddress).symbol()
5         );
```

[I-4] In `TSwapPool::deposit` you have unused variable `poolTokenReserves`, remove it if you do not plan to use it

```
1 -   uint256 poolTokenReserves = i_poolToken.balanceOf(address(this));
```

[I-5] Very important function `TSwapPool::swapExactInput` does not have any natspec

One of the two most important functions to swap tokens, does not have documentation, consider adding one like you do in `swapExactOutput`

```
1
2     // NO DESCRIPTION
3     function swapExactInput(
4         IERC20 inputToken,
5         uint256 inputAmount,
6         IERC20 outputToken,
7         uint256 minOutputAmount,
8         uint64 deadline
9     )
```

[I-6] Function `TSwapPool::swapExactInput` should be external instead of public, because it is not called anywhere in the contract, to be more gas-efficient

```
1 function swapExactInput(IERC20 inputToken, uint256 inputAmount, IERC20
    outputToken, uint256 minOutputAmount, uint64 deadline)
2 -       public revertIfZero(inputAmount) revertIfDeadlinePassed(
    deadline)
3 +       external revertIfZero(inputAmount) revertIfDeadlinePassed(
    deadline)
4         returns (uint256 output)
```